

REMOTE POWER CONTROLLER FOR YOUR COMMODORE-64

Couple a wireless remote and your computer. Instant automation!

Chandler Sowden

■ One of the useful things a home computer can do is control things around the house. Unfortunately, this usually involves stringing control wires from the computer to the device being controlled.

Enter the home wireless remote control system, as manufactured and sold by BSR and Radio Shack. With this system, lights and appliances can be controlled (manually) from a central location without wires.

How it works

The wireless remote modules respond to a pulse




FIG. 1—BLOCK DIAGRAM reveals simplicity of the circuit. Opto-isolators keep AC out of the computer.

coded 125KHz signal placed on the AC power line. The pulses occur in groups of three and are synchronized to the power line. As shown in Figure 1, the device consists of a 60 Hz wave shaping circuit, an




FIG. 2—SCHEMATIC DIAGRAM is simplified because circuit uses computer software to provide many of the necessary functions. Refer to text for in-depth explanation.





FIG. 3—HERE'S THE FOIL PAT-
TERN for those of our readers who
desire to do their own circuit-board
etching.

FIG. 4—...AND WHEN YOU'VE GOT
THE BOARD etched, here's where
all the parts go.



oscillator, a time-out circuit, and optical isolators to keep AC line power out of the computer. Machine language software in the computer first looks at the 60 Hz square wave from the device. This input changes at each zero crossing of the AC power line voltage. The computer then either outputs a group of three pulses or waits until the next zero crossing, depending on the code to be transmitted. These pulse groups go through the optical isolator and time-out circuit and are applied to the oscillator, which places corresponding bursts of 125 KHz on the power line. The grouping of pulses is controlled by the machine language software, and depends on the house code, unit code, etc., to be transmitted. The time-out circuit is to prevent a software hang-up from leaving the oscillator on. If the pulse output from the computer stays high for 2 seconds or more, the oscillator is switched off until the output returns to zero.

The circuit

The circuit is shown in Figure 2. When the computer sends a high logic level signal on the PULSE input, transistor Q1 turns on, which turns on optical isolator IC2 and transistor Q4. This triggers the monostable side of IC1, whose output then releases the RESET input (pin 10) of the astable side, allowing it to oscillate. Its 125 KHz output is amplified by Q5 and then applied to the AC power line through capacitor C4. When the PULSE

input goes low, IC2 turns off and the monostable is reset through resistor R6, thus shutting off the 125 KHz output. If the PULSE input remains high for some reason, the monostable times out after about 2 seconds and shuts off the oscillator. Returning the PULSE input to low then restores normal operation.

Assembly

The power controller can be easily assembled using point-to-point wiring or a printed circuit board. A foil layout is shown in Figure 3, and component placement in Figure 4. Note that the optical isolators are mounted

FIG. 7—PROGRAM LISTING.

```

1 REM   BSR POWER CONTROLLER PROGRAM
2 REM   FOR COMMODORE 64
3 REM   WRITTEN BY G.C. SOWDEN 6/14/85
10 LET P=PEEK(56577)
20 FOR I=1 TO 20
30 IF PEEK(56577)<>P THEN GO TO 100
40 NEXT I
50 PRINT "POWER CONTROLLER IS NOT PLUGGED IN"
60 END
100 POKE 56579,1
102 GOSUB 109
104 GO TO 600
109 RESTORE
110 FOR A=49152 TO 49336
120 READ D
130 POKE A,D
140 NEXT A
142 RETURN
145 REM BSR
150 DATA 169,2,141,42,192

```

```

160 DATA 32,47,192,173,43,192
170 DATA 141,44,192,32,68,192
180 DATA 173,45,192,208,3,32,94,192
190 DATA 173,46,192,141,44,192,32,68
200 DATA 192,206,42,192,208,222,96,0,0
210 DATA 0,105,0,0,171
220 REM SC
230 DATA 32,94,192,169,3
240 DATA 141,67,192,32,120,192
250 DATA 206,67,192,208,248,32,94,192
260 DATA 96,0
270 REM BITS
280 DATA 169,8,141,93,192,14,44,192
290 DATA 144,6,32,120,192,76,87,192
300 DATA 32,94,192,206,93,192,208,237
310 DATA 96,0
320 REM TZ
330 DATA 169,0,141,119,192,206,119,192
340 DATA 208,251
350 DATA 173,1,221,141,119,192
360 DATA 173,1,221,205,119,192,240,248
370 DATA 96,0
380 REM T1
390 DATA 169,3,141,160,192,173,161,192
400 DATA 9,1,141,1,221,162,186,202
410 DATA 208,253,173,161,192,141,1,221
420 DATA 169,183,141,162,192
430 DATA 206,162,192,208,251
440 DATA 206,160,192,208,222,96,0,0,0
450 REM UNIT CODES
460 DATA 105,169,89,153,86,150,102,166
470 DATA 106,170,90,154,85,149,101,165
480 DATA 179,181,173,171,205,203
600 LET N=100
610 DIM T(N),U(N),F(N),S$(16)
620 FOR I=1 TO N
630 READ T(I),U(I),F(I)
635 IF T(I)>2359 THEN GO TO 1000
640 NEXT I
650 DATA 1201,1,1
651 DATA 1202,1,2
999 DATA 9999,0,0
1000 LET C$="A"
1010 PRINT "TIME (HHMM)"
1020 INPUT T$
1030 LET T1$=T$+"00"
1040 LET T$=LEFT$(T1$,4)
1060 LET H$=LEFT$(T$,2)
1070 LET M$=RIGHT$(T$,2)
1090 LET T1=VAL(T$)
1100 FOR I=1 TO N
1105 IF T(I)>2359 THEN GO TO 1160
1110 IF T(I)<>T1 THEN GO TO 1150
1120 LET U1=U(I)
1130 LET F1=F(I)
1140 GOSUB 8000
1150 NEXT I
1160 GOSUB 7000
1170 LET T0$=LEFT$(T1$,4)
1180 IF PEEK(197)<>64 THEN GO TO 1220
1190 IF T0$=T$ THEN GO TO 1170
1200 LET T$=T0$
1210 GO TO 1060
1220 IF PEEK(197)<>64 THEN GO TO 1220
1230 INPUT D$
1240 INPUT "TYPE DEVICE NUMBER: ";U1
1250 IF U1=0 THEN END
1260 INPUT "FUNCTION?(TYPE 1 FOR ON, 2 FOR OFF)";F1
1270 GOSUB 8000
1280 GO TO 1160
7000 PRINT "CLR:"
7010 PRINT H$;" ";M$
7020 PRINT
7030 PRINT "DEVICE NUMBER AND STATUS:"
7040 PRINT
7050 FOR K=1 TO 16
7060 PRINT S$(K);
7065 NEXT K
7070 PRINT
7080 PRINT "PRESS RETURN FOR MANUAL CONTROL"
7090 RETURN
8000 LET S1$=" OFF"
8030 IF F1=1 THEN LET S1$=" ON "
8040 LET S$(U1)=STR$(U1)+S1$
8050 GOSUB 9000
8060 RETURN
9000 GOSUB 109
9002 POKE 49195,PEEK(49250)+ASC(C$)
9008 POKE 49198,PEEK(49314)+U1
9010 POKE 49197,1
9020 SYS 49152
9030 FOR J=1 TO 10

```

```

9031 NEXT J
9032 IF F1<5 THEN GO TO 9050
9034 LET F2=2*(F1-100*INT(F1/100))
9036 LET F1=INT(F1/100)
9040 POKE 49153,F2
9050 POKE 49198,PEEK(49330)+F1
9060 POKE 49197,0
9070 SYS 49152
9075 POKE 49153,2
9080 PRINT U1,F1
9082 FOR J=1 TO 200
9084 NEXT J
9090 RETURN

```




FIG. 5—WIRING FOR THE CONNECTOR is shown above. Provide good mechanical connections for all leads and then carefully solder.




FIG. 6—DIMENSIONS ARE PROVIDED here for openings in the connector. See text for additional explanation.

facing opposite directions. You'll have to cut openings for the AC cord and the computer card-edge connector. The connector mounts on the side of the box, and it then plugs into the Commodore User Port.

After the components are in place, install the power cord. Solder hookup wire to the card-edge connector, as shown in figure 5. Cut a rectangular opening in the



FIG. 8—THE CIRCUIT BOARD completely wired, ready for installation in the box.




FIG. 9—POWER CONTROLLER installed in the box with the connector attached. Note that only four of the 24 pins are used.

plastic box for the card-edge connector, as shown in Figure 6.

Using a volt-ohmmeter, measure the resistance between a blade on the power plug and the connector contacts. There should be no continuity between the power plug and any contact on the connector. Next, plug the unit into a wall outlet and CAREFULLY measure the DC voltage at the output of the bridge rectifier. It should be about 8 volts. Be careful, because most of the components have power line voltage on them.

The software

Type the software in from the listing given in Figure 7. Double check the listing, especially the DATA statements. Save the program on tape or disk before you run the program.

The program is designed as a clock-time controller, but it also allows manual control of the remote modules from the computer keyboard. The program allows up to 100 on or off operations in a day, but this could be changed by changing line 600. The desired on/off times are in DATA statements is: (line number) DATA time, unit, code where "code" is a 1 for on, 2 for off. In the program, line 650 means "At 12:01 (24 hour time), turn unit 1, ON (code 1)," while line 651 says "At 12:02, turn unit 1, OFF (code 2)."

The "house code" for the power controller is set at line 1000. Sixteen different house codes are provided,




FIG. 10—POWER CONTROLLER plugs into the user port at the rear of the cabinet.




FIG. 11—THIS ILLUSTRATES the power controller connected to the computer. Be sure the connector location is such that the box does not cover the opening for the cassette interface.

in case your neighbor uses a similar system.

Take a remote module, set it to House Code "A" and unit code "1." Plug it in and plug a lamp into it. Run the program. It will then ask for the time. Type in the time in 4-digit, 24-hour format, then return. The program will clear the screen, print a heading, then a line which asks you to "press return for manual control." Press RETURN. It will say "type device number;" type 1 (and return); it

PARTS LIST

Semiconductors

- BR1—1 ampere, 50V. bridge rectifier
- IC1—LM556 dual timer
- IC2, IC3—TIL111 (or similar) optical isolator
- Q1-Q4—2N3904 or similar transistor
- Q5—TIP31 transistor

Resistors

(All resistors 1/4 watt unless otherwise indicated)

- R1—10,000-ohm
- R2—4700-ohm
- R3—150-ohm
- R4—470-ohm
- R5—15,000-ohm
- R6-R8, R10—1000-ohm
- R9—1-megohm
- R11—6800-ohm
- R12—10,000-ohm 1/2 watt potentiometer
- R13—220-ohm
- R14—100,000-ohm
- R15—10-ohm, 2 watts

Capacitors

(All capacitors at least 16V.)

- C1—.0047 μ F.
- C2—3.3 μ F. C3—470pF
- C4—.22 μ F.
- C5—1000 μ F.

Miscellaneous parts and equipment

- F1—1/4 a. fuse (see text)
- J1—Card-edge connector, Texas Instruments No. H411121-12 (available from Digi-Key Corp., P. O. Box 577, Thief River Falls, MN 56701-9988)
- T1—6.3V 300mA transformer
- Line cord, box, etc.

will ask "Function? (Type 1 for on, 2 for off)." Type 1 (and return). The lamp should turn on. If it doesn't, try checking the house and unit codes, then check the program for typos. If that doesn't help, then try tweaking potentiometer R12. ◀▶